

Berechnung einer Cosinus-Tabelle mit Hilfe des Additionstheorems


Mathematik-Referat, Marc Antoni, Ohm-Gymnasium Klasse 10c, Juni 2001

Inhalt

1. Verdeutlichung des Berechnungsprinzips anhand von Grafiken
2. Ableitung des iterativen (auf Wiederholung beruhenden) Algorithmus aus dem Additionstheorem für den Cosinus
3. Flußdiagramm für das Computerprogramm
4. Quelltext des QBasic-Programms mit Erklärung der Befehle
5. Vorführung des Programms mit Beamer

1. Verdeutlichung des Funktionsprinzips anhand von Grafiken

Schritt 1: Ermittlung des cos eines kleinen Winkels $\cos(60^\circ/2^3)$



$$\cos \frac{a}{2} = \sqrt{\frac{1 + \cos a}{2}}$$

Ableitung: siehe Kap. 2


$$\cos 60^\circ = 0,5 \quad (\text{siehe Kap. 2})$$

$$\cos 30^\circ = \sqrt{\frac{1 + 0,5}{2}} = 0,866$$

$$\cos 15^\circ = \sqrt{\frac{1 + 0,866}{2}} = 0,966$$

$$\cos 7,5^\circ = \sqrt{\frac{1 + 0,966}{2}} = 0,991$$

Schritt 2: Ermittlung des cos vom Vielfachen dieses Winkels $\cos(n \times 7,5^\circ)$



Ableitung: siehe Kap. 2

$$\cos(a + b) = \cos a \cdot \cos b - \sqrt{(1 - (\cos a)^2) \cdot (1 - (\cos b)^2)}$$

$$k\text{-ter Iterationsschritt: } \cos(a_k) = \cos(a_{k-1} + 7,5^\circ) = \cos a_{k-1} \cdot 0,991 - \sqrt{(1 - (\cos a_{k-1})^2) \cdot (1 - 0,991^2)}$$


α_k = aktueller Winkel

α_{k-1} = Winkel des letzten Iterationsschrittes

2. Ableitung der in Kap. 1 verwendeten Formeln

1. Ermittlung von $\cos 60^\circ$

Über das gleichseitige Dreieck: $\cos 60^\circ = \frac{a/2}{a} = \frac{1}{2}$




Ableitung einer Funktion der Form $\cos \frac{a}{2} = f(\cos a)$

$$\cos(b + g) = \cos b \cdot \cos g - \sin b \cdot \sin g \quad (\text{Additionstheorem}) \quad (1)$$

$$b = a/2 \quad ; \quad g = a/2 \quad (2)$$

$$(2 \text{ in } 1) \rightarrow \cos a = \left(\cos \frac{a}{2}\right)^2 - \left(\sin \frac{a}{2}\right)^2 \quad (3)$$

$$1 = \left(\cos \frac{a}{2}\right)^2 + \left(\sin \frac{a}{2}\right)^2 \Rightarrow \left(\sin \frac{a}{2}\right)^2 = 1 - \left(\cos \frac{a}{2}\right)^2 \quad (4)$$



Satz des Pythagoras am Einheitskreis

$$(4 \text{ in } 3) \rightarrow \cos a = 2\left(\cos \frac{a}{2}\right)^2 - 1$$

$$\left(\cos \frac{a}{2}\right)^2 = \frac{1 + \cos a}{2}$$

$$\boxed{\cos \frac{a}{2} = \sqrt{\frac{1 + \cos a}{2}}}$$

2. Ableitung einer Funktion der Form $\cos(a + b) = f(\cos a, \cos b)$


$$\cos(a + b) = \cos a \cdot \cos b - \sin a \cdot \sin b \quad (\text{Additionstheorem}) \quad (1)$$

$$(\sin a)^2 + (\cos a)^2 = 1 \rightarrow \sin a = \sqrt{1 - (\cos a)^2} \quad (2)$$

(Satz des Pythagoras am Einheitskreis; siehe oben)

$$(2 \text{ in } 1) \rightarrow \boxed{\cos(a + b) = \cos a \cdot \cos b - \sqrt{1 - (\cos a)^2} \cdot \sqrt{1 - (\cos b)^2}}$$

3. Flussdiagramm



4. Quelltext des QBasic-Programms mit Erklärung der Befehle

```

*****
COSINTABE.BAS - Berechnung einer Cosinus-Tabelle über das Additionstheorem
=====
Dieses QBasic-Programm berechnet eine Cosinus-Tabelle, ohne die - in QBasic
verfügbare - Cos-Funktion selbst zu verwenden. Es werden iterative Algorith-
men benutzt, die ausschließlich auf Quadrat- und Wurzelfunktionen basieren.
Die Berechnung läuft in 2 Teilen ab.

In Teil 1 berechnet das Programm den minimalen Winkel, der dann als Schritt-
weite für Teil 2 dient. Dieser Winkel ergibt sich aus der fortwährenden
(iterativen) Halbierung des Winkels 60°, dessen Cosinus bekanntlich den Wert
0,5 hat. Die Cosinusfunktion eines halbierten Winkels ergibt sich über eine
Formel aus dem Cosinus des ganzen Winkels, die aus dem Additionstheorem
ableitbar ist. Die Anzahl "nmax" der Halbierungsschritte wird vom Anwender
eingegeben. Der Minimalwinkel "alphamin" beträgt also  $60^\circ/2^{nmax}$  .

In Teil 2 wird der Cosinus von allen Vielfachen des Minimalwinkels "aphamin"
im Bereich 0 bis 90° berechnet. Hierbei kommt ebenfalls das Additionstheorem
zur Anwendung.

(c) Marc Antoni, 28.06.02
Mailto:marc@antonis.de
http://www.antonis.purespace.de

Dies Programm ist downloadbar unter
http://www.antonis.purespace.de/cosintab.htn

*****

COLOR 0, 7 'Schwarze Schrift auf grauem Grund
WIDTH 80, 50 'Bildschirmgröße 80 Spalten, 50 Zeilen
DO
CLS

'----- Teil 1 - Ermittlung des Minimalwinkels und dessen cos -----
PRINT " ===== Anzeige einer Cosinus-Tabelle ====="
PRINT " Der minimale Winkel (Schrittweite) beträgt  $60^\circ/2^n$ "
INPUT " Gib die Winkelauflösung n ein"; nmax%
c# = .5

FOR n% = 1 TO nmax% 'Schleife über die Iterationsschritte
    c# = SQR((1 + c#) / 2) 'Berechnung des cos des halben Winkels
NEXT n% 'über die Quadratwurzel ("Square Root")

alphamin# = 60 / (2 ^ nmax%)
PRINT
PRINT " cos 0°", "      = 1.0000000000000000" 'Anzeige von cos 0°
PRINT " cos"; alphamin#; "°", "      = "; c# 'Anzeige von cos (alphamin)

'----- Teil 2 - Ermittlung u. Anzeige der cos-Werte von Vielfachen -----
'----- des Minimalwinkels alphamin im Bereich 0...90° -----
d# = c# 'Anfangswert für die Iteration =
'cos von alphamin
kmax% = 90 / alphamin# 'Anzahl der Iterationsschritte =
'90°/Schrittweite

FOR k% = 2 TO kmax% 'Schleife über die Iterationsschritte
    d# = d# * c# - SQR((1 - d# ^ 2) * (1 - c# ^ 2))
    PRINT " cos"; k% * alphamin#; "°";
    LOCATE , 20 'Cursor auf Spalte 29
PRINT "= "; d#
NEXT k%

'----- Wiederholen/ Beenden-Dialog -----
PRINT 'Leerzeile anzeigen
PRINT " Weiter mit beliebiger Taste, Abbruch mit Esc"
DO: taste$ = INKEY$: LOOP WHILE taste$ = "" 'Warten auf Tastenbetätigung
LOOP WHILE taste$ <> CHR$(27) 'Neue Eingabe, wenn kein Esc
END 'Beenden mit Esc-Taste

```